```
> restart: with(linalg):with(plots):with(liesymm):with(difforms):
  setup(x,y,z,t):
  defform(t=0,s=0,Z=0,phi=0,theta=0,r=0,p=const,q=const,a=const,Y=0,
  X=0,E=0,U=0,V=0,W=0,x=0,y=0,z=0,CL=const,beta=0,gamma=0,Omega=0,C=
  0,Vz=const,kappa=const,omega=const,Q=const,R=const);
Warning, new definition for norm
Warning, new definition for trace
Warning, new definition for close
Warning, new definition for `&^`
Warning, new definition for d
Warning, new definition for mixpar
Warning, new definition for wdegree
```

alors.mws

R. M. Kiehn
rkiehn2352@aol.com
last update 01/25/2000

# Lorentz transformations

Lorentz transformations: **x** => **X** = [L] **x** ,
take the indefinite quadratic form x^2+y^2+z^2-C^2t^2
into the indefinite quadratic form. X^2+Y^2+Z^2-C^2T^2
via the matrix product <**x**|[Minkowski]|**x**>.
[Minkowski] is the Minkowski metric designated as [MM] in that which follows.
The Lorentz transformations are automorphisms of the correlation group relative to [MM]

**(transpose[L]) [MM] [L] = [MM]**

Lorentz transformations preserve the Minkowski metric.
Applications to electromagnetism and the concept of signals are described at
http://www22.pair.com/csdc/pdf/specrela.pdf
and
http://www22.pair.com/csdc/pdf/specrel2.pdf

```
> dim:=4;coord :=[x,y,z,Ct];DR:=d(coord);
```

$$dim := 4$$

$$coord := [x, y, z, Ct]$$

$$DR := [d(x), d(y), d(z), d(Ct)]$$

```
> MM:=array([[1,0,0,0],[0,1,0,0],[0,0,1,0],[0,0,0,-1]]);
```

$$MM := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 \end{bmatrix}$$

MM is the Minkowski metric relative to the coordinates, [x,y,z,t]

# Proper and Improper Lorentz Transformations.

Lorentz transformations can be represented by those invertible matrices that preserve the indefinite quadratic form ,

**(transpose[L]) [MM] [L] = [MM]**

As the matrices are invertible their determinant must not be zero. The collection of all such matrices form two disjoint components, the proper and the improper Lorentz transformations of the Lorentz group. Without being too precise, the proper Lorentz transformations have qualities that are similar to euclidean rotations, and the improper Lorentz transformations have qualities that are similar to euclidean rotations.

Proper Lorentz transformations can be represented by matrices that have a determinant > 0; .hence they include the Identity transformation. The proper Lorentz transformations form a subgroup of the full Lorentz group.

Improper Lorentz transformations have a determinant < 0. The improper Lorentz transformations do not form a subgroup, for they do not include the identity transformation (whose determinant is plus 1).

Special Lorentz transformations have determinant equal to plus 1, hence preserve the volume measure.

Special improper Lorentz transformations have determinant equal to minus 1.

Lorentz transformations can be represented by 4x4 matrices, and each matrix can be constructed from products of primitive matrix generators. The primitive generators depend upon a single arbitrary function (in the sense of a parameter), and correlate pairs of coordinate variables. The space-space pairs (xy, yz, zx) have generators that are similar to the rotation and reflection matrices of euclidean space. The space-time pairs (xt,yt,zt) have similar generators, but the correlation behavior of the coordinate pairs is only approximated by the rotations and reflections of objects in euclidean space when the the generating parameter has values close to zero..

# Proper Lorentz space-time transformations

First consider a matrix generator of a Lorentz transformation that is associated with the xt pair of coordinates. The functional arguement or "generator parameter", beta, is often identified with the ratio of the velocity along the x axis to the speed of light. The function beta can be either positive or negative, which is interpreted by saying that the direction of the velocity changes sign. To take this into account, the parameter beta(x,y,z,t) will be multiplied by the arbitrary number Q, where Q will take on the values of either plus or minus one. The idea is similar to the notion in electromagnetism, where the direction of the current is related to the sign of the charge. Herein Q will be defined as the "charge conjugation" operator, or the operator that "changes the direction of velocity. In addition, a numeric factor R (with values plus or minus 1) can be introduced to account for "simultaneous reflections" in both coordinates (for small beta).

The transformation for R = + 1 can be interpreted (at least for small beta) as T ~ + t , X ~ + x"

The transformation for R = -1 can be interpreted (at least for small beta) as a "reflection in both the T and the X axis.

X ~ -x, T ~ - t" . R and Q are not the same.

As an example, consider the matrix defined (to within a factor) as:

**Lxt**

```
> Lxt := evalm(matrix([[R/(1-beta^2)^(1/2), 0, 0,
```

```
Q*beta/(1-beta^2)^(1/2)], [0, 1, 0, 0], [0, 0, 1,0],
[Q*beta/(1-beta^2)^(1/2),0,0,
R/(1-beta^2)^(1/2)]]));DETLxt:=subs(Q^2=1,R^2=1,det(Lxt));
```

$$Lxt := \begin{bmatrix} \frac{R}{\sqrt{1-\beta^2}} & 0 & 0 & \frac{Q\,\beta}{\sqrt{1-\beta^2}} \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ \frac{Q\,\beta}{\sqrt{1-\beta^2}} & 0 & 0 & \frac{R}{\sqrt{1-\beta^2}} \end{bmatrix}$$

$$DETLxt := 1$$

Note that the determinant is plus one, so these "proper" transformations include the identity transformation. For small beta, note that x~X and t~T, so that the "signs" of the coordinates stay the same. Yet there is a difference between the two generators for the two choices of Q. In fact [Lxt(Q = +1)] is the inverse matrix to [Lxt(Q = -1)]. Velocity direction conjugation (plus Q goes to minus Q) forms the inverse Lorentz matix of the proper Lorentz space-time generator.

***

To prove that the Minkowski metric is an invariant of the correlation transformation generated by the above matrix: compute (transpose[L]) [MM] [L] = [MM]

```
> `MM'`:=(subs(Q^2=1, R^2 =
  1,simplify(innerprod(transpose(Lxt),MM,Lxt))));
```

$$MM' := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 \end{bmatrix}$$

To prove that Q = minus 1 is the inverse of Q = plus 1, compute the product of LxtplusQ times LxtminusQ:

```
> LxtplusQ:=(subs(Q=1,evalm(Lxt))):LxtminusQ:=(subs(Q=-1,evalm(Lxt))
  ):
> simplify(subs(R^2=1,innerprod(LxtminusQ,LxtplusQ)));
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

QED

The results are valid if Q plus goes to Q minus ( Q minus is interpreted by saying the velocity direction has reversed). Note that Q plus goes to Q minus does not imply that t goes to minus T, or x goes to minus X.

# Improper space-time Lorentz transformations

The improper Lorentz transformations have a negative determinant, and although are elements of a separate component of the group of Lorentz transformations, they, as a components, do not form a group (no identity element has det = minus 1).

As for the proper Lorentz transformations, it is useful to introduce another number R, (similar to Q) that takes on values of either plus or minus 1. The generator of improper Lorentz transformations involving the xt pair of coordinates takes on the form, **Lxtimp:**

```
> Lxtimp := matrix([[R/(1-beta^2)^(1/2), 0, 0,
  Q*beta/(1-beta^2)^(1/2)], [0, 1, 0, 0], [0, 0, 1,0],
  [-Q*beta/(1-beta^2)^(1/2),0,0,
  -R/(1-beta^2)^(1/2)]]);DETLxtimp:=subs(Q^2=1,R^2=1,det(Lxtimp));
>
```

$$Lxtimp := \begin{bmatrix} \frac{R}{\sqrt{1-\beta^2}} & 0 & 0 & \frac{Q\,\beta}{\sqrt{1-\beta^2}} \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ -\frac{Q\,\beta}{\sqrt{1-\beta^2}} & 0 & 0 & -\frac{R}{\sqrt{1-\beta^2}} \end{bmatrix}$$

$$DETLxtimp := -1$$

To prove that the improper generators are Lorentz transformations, compute the quadratic form:

```
> `MM'`:=(subs(Q^2=1,R^2=1,simplify(innerprod(transpose(Lxtimp),MM,Lxtimp))));
```

$$MM' := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 \end{bmatrix}$$

QED

The transformation for R = + 1 can be interpreted (at least for small beta) as a "reflection in the T axis. T ~ - t , X ~ + x"

The transformation for R = -1 can be interpreted (at least for small beta) as a "reflection in the X axis. X ~ -x, T ~ + t"

****

The remarkable result is that these 4 distinct improper coordinate reflections as Lorentz generators are such that each matrix is its own inverse. (Proof below)

Matrices such that the square of the matrix is the identity are said to be matrices **"in involution"**.

Another term is "a reflexive matrix". Reflexive matrices define a "symmetry" or an oscillation

*This idea is not the same as the symmetrical and anti-symmetrical parts of the matrix. (The symmetrical projector is {Lxtr + transpose(Lxtr)}/2 and the antisymmetrical projector is {Lxtr - transpose(Lxtr)}/2.)*

Further note that beta goes to minus beta does not generate the inverse matrix.

**

To prove involution, compute:

```
> subs(Q^2=1,R^2=1,innerprod((Lxtimp),Lxtimp));
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

QED for any of the 4 combinations of R and Q

Question for later study:

**How do these 4 improper generators relate (if at all) to spinors???**

The format for other generators of other coordinate pairs can be constructed by inspection:

The Q R and beta functions are distinct for each generator format. For the yt pair, beta is often interpreted as the ratio of the velocity in they direction to the speed of light; etc.

```
> Lyt := evalm(matrix([[0, 1, 0, 0],[0,R/(1-beta^2)^(1/2), 0,
  Q*beta/(1-beta^2)^(1/2)], [0, 0, 1,0],
  [0,Q*beta/(1-beta^2)^(1/2),0,
  R/(1-beta^2)^(1/2)]]));DETLxt:=subs(Q^2=1,R^2=1,det(Lxt));
```

$$Lyt := \begin{bmatrix} 0 & 1 & 0 & 0 \\ 0 & \frac{R}{\sqrt{1-\beta^2}} & 0 & \frac{Q\,\beta}{\sqrt{1-\beta^2}} \\ 0 & 0 & 1 & 0 \\ 0 & \frac{Q\,\beta}{\sqrt{1-\beta^2}} & 0 & \frac{R}{\sqrt{1-\beta^2}} \end{bmatrix}$$

$$DETLxt := 1$$

```
> Lzt := evalm(matrix([ [0, 1, 0, 0], [0, 0, 1,0],
  [0,0,R/(1-beta^2)^(1/2),
  Q*beta/(1-beta^2)^(1/2)],[0,0,Q*beta/(1-beta^2)^(1/2),
  R/(1-beta^2)^(1/2)]]));DETLxt:=subs(Q^2=1,R^2=1,det(Lxt));
```

$$Lzt := \begin{bmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & \frac{R}{\sqrt{1-\beta^2}} & \frac{Q\,\beta}{\sqrt{1-\beta^2}} \\ 0 & 0 & \frac{Q\,\beta}{\sqrt{1-\beta^2}} & \frac{R}{\sqrt{1-\beta^2}} \end{bmatrix}$$

$$DETLxt := 1$$

The format of the improper Lorentz tranformations for other coodinate pairs can be constructed in a similar manner.

# Compound space-time Lorentz Transformations

The idea is to generate other Lorentz transformation matrices by multiplying generators in a matrix fashion. The Lorentz generators do not commute. Hence the order of application is important and sometimes leads to surprising results.

The first example is to demonstrate that the "so called" velocity addition formula. Multiply two proper generators for the same xt pair. Lxtb2 times Lxtb2. First assume the two betas are the same and compute Lxtb1 times Lxtb1:

```
> Lxtb1 := evalm(matrix([[R1/(1-beta1^2)^(1/2), 0, 0,
  Q1*beta1/(1-beta1^2)^(1/2)], [0, 1, 0, 0], [0, 0, 1,0],
  [Q1*beta1/(1-beta1^2)^(1/2),0,0,
  R1/(1-beta1^2)^(1/2)]]));DETLxtb1:=simplify(subs(Q1^2=1,R1^1=1,det
  (Lxtb1)));
```

$$Lxtb1 := \begin{bmatrix} \frac{R1}{\sqrt{1-\beta1^2}} & 0 & 0 & \frac{Q1\,\beta1}{\sqrt{1-\beta1^2}} \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ \frac{Q1\,\beta1}{\sqrt{1-\beta1^2}} & 0 & 0 & \frac{R1}{\sqrt{1-\beta1^2}} \end{bmatrix}$$

$$DETLxtb1 := 1$$

```
> Lxtsq:=subs(Q1^1=1,R1^2=1,innerprod(Lxtb1,Lxtb1));simplify(subs(R1
  ^2=1,det(Lxtsq)));
```

$$Lxtsq := \begin{bmatrix} -\frac{1+\beta1^2}{-1+\beta1^2} & 0 & 0 & -2\frac{R1\,\beta1}{-1+\beta1^2} \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ -2\frac{R1\,\beta1}{-1+\beta1^2} & 0 & 0 & -\frac{1+\beta1^2}{-1+\beta1^2} \end{bmatrix}$$

$$1$$

Show that the compound matrix is a Lorentz matrix:

```
> `MM'`:=(simplify(subs(R1^2=1,innerprod(transpose(Lxtsq),MM,Lxtsq))
  ));
```

$$MM' := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 \end{bmatrix}$$

QED

Now use two different velocity ratios, beta1 and beta2:

```
> Lxtb2 := evalm(matrix([[R2/(1-beta2^2)^(1/2), 0, 0,
  Q2*beta2/(1-beta2^2)^(1/2)], [0, 1, 0, 0], [0, 0, 1,0],
  [Q2*beta2/(1-beta2^2)^(1/2),0,0,
  R2/(1-beta2^2)^(1/2)]]));DETLxtb2:=simplify(subs(Q2^2=1,R2^2=1,det
  (Lxtb2)));
```

$$Lxtb2 := \begin{bmatrix} \frac{R2}{\sqrt{1-\beta 2^2}} & 0 & 0 & \frac{Q2\,\beta 2}{\sqrt{1-\beta 2^2}} \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ \frac{Q2\,\beta 2}{\sqrt{1-\beta 2^2}} & 0 & 0 & \frac{R2}{\sqrt{1-\beta 2^2}} \end{bmatrix}$$

$$DETLxtb2 := 1$$

Create the product

```
> Prodb1b2:=innerprod(Lxtb2,Lxtb1);
```

$$Prodb1b2 := \begin{bmatrix} \frac{R2\,R1 + Q2\,\beta 2\,Q1\,\beta 1}{\sqrt{1-\beta 2^2}\,\sqrt{1-\beta 1^2}} & 0 & 0 & \frac{R2\,Q1\,\beta 1 + Q2\,\beta 2\,R1}{\sqrt{1-\beta 2^2}\,\sqrt{1-\beta 1^2}} \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ \frac{R2\,Q1\,\beta 1 + Q2\,\beta 2\,R1}{\sqrt{1-\beta 2^2}\,\sqrt{1-\beta 1^2}} & 0 & 0 & \frac{R2\,R1 + Q2\,\beta 2\,Q1\,\beta 1}{\sqrt{1-\beta 2^2}\,\sqrt{1-\beta 1^2}} \end{bmatrix}$$

Note the sensitivity on the Reflection and velocity direction numbers. R1,R2,Q1,Q2
The usual velocity composition for velocities in the "same" direction is determined from the expression:
R1=R2=Q1=Q2 = 1

```
> 
> `Velocity composition
  rule`:=subs(Q1=1,R1=1,Q2=1,R2=1,evalm(Prodb1b2));
```

$$Velocity\ composition\ rule := \begin{bmatrix} \frac{1 + \beta 2\,\beta 1}{\sqrt{1-\beta 2^2}\,\sqrt{1-\beta 1^2}} & 0 & 0 & \frac{\beta 1 + \beta 2}{\sqrt{1-\beta 2^2}\,\sqrt{1-\beta 1^2}} \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ \frac{\beta 1 + \beta 2}{\sqrt{1-\beta 2^2}\,\sqrt{1-\beta 1^2}} & 0 & 0 & \frac{1 + \beta 2\,\beta 1}{\sqrt{1-\beta 2^2}\,\sqrt{1-\beta 1^2}} \end{bmatrix}$$

The idea is that Lxt(beta1) times Lxt(beta2) does not equal Lxt(beta1 + beta2). In Lorentz transformations the "velocity" vector is not additive in standard sense.

Now repeat the same multiplication of two improper Lorentz transformations.
**A self product of improper generators yields the identity, as the improper generator matrices are reflexive.**
But consider different Reflections and different values of Q for the same xt pair:

```
> Lxtb1imp := evalm(matrix([[R1/(1-beta1^2)^(1/2), 0, 0,
  -Q1*beta1/(1-beta1^2)^(1/2)], [0, 1, 0, 0], [0, 0, 1,0],
  [Q1*beta1/(1-beta1^2)^(1/2),0,0,
  -R1/(1-beta1^2)^(1/2)]]));DETLxtb1imp:=simplify(subs(Q1^2=1,R1^2=1
  ,det(Lxtb1imp)));
```

$$Lxtb1imp := \begin{bmatrix} \frac{R1}{\sqrt{1-\beta1^2}} & 0 & 0 & -\frac{Q1\,\beta1}{\sqrt{1-\beta1^2}} \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ \frac{Q1\,\beta1}{\sqrt{1-\beta1^2}} & 0 & 0 & -\frac{R1}{\sqrt{1-\beta1^2}} \end{bmatrix}$$

$$DETLxtb1imp := -1$$

```
> Lxtb2imp := evalm(matrix([[R2/(1-beta2^2)^(1/2), 0, 0,
-Q2*beta2/(1-beta2^2)^(1/2)], [0, 1, 0, 0], [0, 0, 1,0],
[Q2*beta2/(1-beta2^2)^(1/2),0,0,
-R2/(1-beta2^2)^(1/2)]]));DETLxtb2imp:=(subs(Q2^2=1,R2^2=1,det(Lxt
b2imp)));
```

$$Lxtb2imp := \begin{bmatrix} \frac{R2}{\sqrt{1-\beta2^2}} & 0 & 0 & -\frac{Q2\,\beta2}{\sqrt{1-\beta2^2}} \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ \frac{Q2\,\beta2}{\sqrt{1-\beta2^2}} & 0 & 0 & -\frac{R2}{\sqrt{1-\beta2^2}} \end{bmatrix}$$

$$DETLxtb2imp := -1$$

Choose beta1=beta2

```
> prodb1b2imp:=subs(R2^2=1,R1^2=1,Q1^2=1,Q2^2=1,beta2=beta1,innerpro
d(Lxtb2imp,Lxtb1imp));`MM'`:=simplify(subs(R2^2=1,R1^2=1,Q1^2=1,Q2
^2=1,beta2=beta1,innerprod(prodb1b2imp,MM,prodb1b2imp)));
>
```

$$prodb1b2imp := \begin{bmatrix} -\frac{-R2\,R1+Q2\,\beta1^2\,Q1}{1-\beta1^2} & 0 & 0 & -\frac{R2\,Q1\,\beta1-Q2\,\beta1\,R1}{1-\beta1^2} \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ -\frac{R2\,Q1\,\beta1-Q2\,\beta1\,R1}{1-\beta1^2} & 0 & 0 & -\frac{-R2\,R1+Q2\,\beta1^2\,Q1}{1-\beta1^2} \end{bmatrix}$$

$$MM' := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 \end{bmatrix}$$

The resulting product of xt pair generators (for the same beta) depends on the values of R and Q. The products are in involution if R1=R2 and Q1= Q2 or R1= -R1 and Q1=-Q2.

```
> Z:=subs(R1^2=1,Q1^2=1,R2^2=1,Q2^2=1,evalm(innerprod(prodb1b2imp,pr
odb1b2imp))):ZZ:=subs(R2=R1,Q2=Q1,evalm(Z)):`Involution (for R2=R1
and Q1=Q2) or (R2=-R1 and Q2=-Q1)
```

```
`=simplify(subs(R1^2=1,Q1^2=1,evalm(ZZ)));
```

Lxtimp(b1) times Lxtimp(b1) = I (no velocity addition law!!!)

\>

$$\textit{Involution (for R2=R1 and Q1=Q2) or (R2=-R1 and Q2=-Q1)} = \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

If Q2=-Q1 and R2=+R1 or Q2=+Q1 and R2= - R1, then the product does not yield the identity

```
> Z:=subs(R1^2=1,Q1^2=1,R2^2=1,Q2^2=1,evalm(innerprod(prodb1b2imp,prodb1b2imp))):ZZ:=subs(R2=-R1,Q2=Q1,evalm(Z)):`Involution (for R2=-R1 and Q1=Q2) or (R2=R1 and Q2=-Q1) `=simplify(subs(R1^2=1,Q1^2=1,evalm(ZZ)));
```

$$\textit{Involution (for R2=-R1 and Q1=Q2) or (R2=R1 and Q2=-Q1)} =$$

$$\begin{bmatrix} \frac{1+6\,\beta1^2+\beta1^4}{(-1+\beta1^2)^2} & 0 & 0 & -4\,\frac{(1+\beta1^2)\,\beta1\,R1\,Q1}{(-1+\beta1^2)^2} \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ -4\,\frac{(1+\beta1^2)\,\beta1\,R1\,Q1}{(-1+\beta1^2)^2} & 0 & 0 & \frac{1+6\,\beta1^2+\beta1^4}{(-1+\beta1^2)^2} \end{bmatrix}$$

# Compound space time Lorentz transformations for different axis pairs

now study the round trip Lyt(-beta) times Lxt(-beta) times Lyt(beta) times Lxt(beta)

```
> LxtplusQ:=(subs(Q=1,beta=beta1,R=1,evalm(Lxt)));LxtminusQ:=(subs(Q=-1,beta=beta1,R=1,evalm(Lxt)));
> LytplusQ:=(subs(Q=1,beta=beta2,R=1,evalm(Lyt)));LytminusQ:=(subs(Q=-1,beta=beta2,R=1,evalm(Lyt)));
> Roundtrip:=simplify(subs(R^2=R2,beta1=beta2,innerprod(LytminusQ,LxtminusQ,LytplusQ,LxtplusQ)));
```

$$\mathit{LxtplusQ} := \begin{bmatrix} \frac{1}{\sqrt{1-\beta1^2}} & 0 & 0 & \frac{\beta1}{\sqrt{1-\beta1^2}} \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ \frac{\beta1}{\sqrt{1-\beta1^2}} & 0 & 0 & \frac{1}{\sqrt{1-\beta1^2}} \end{bmatrix}$$

$$LxtminusQ := \begin{bmatrix} \frac{1}{\sqrt{1-\beta 1^2}} & 0 & 0 & -\frac{\beta 1}{\sqrt{1-\beta 1^2}} \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ -\frac{\beta 1}{\sqrt{1-\beta 1^2}} & 0 & 0 & \frac{1}{\sqrt{1-\beta 1^2}} \end{bmatrix}$$

$$LytplusQ := \begin{bmatrix} 0 & 1 & 0 & 0 \\ 0 & \frac{1}{\sqrt{1-\beta 2^2}} & 0 & \frac{\beta 2}{\sqrt{1-\beta 2^2}} \\ 0 & 0 & 1 & 0 \\ 0 & \frac{\beta 2}{\sqrt{1-\beta 2^2}} & 0 & \frac{1}{\sqrt{1-\beta 2^2}} \end{bmatrix}$$

$$LytminusQ := \begin{bmatrix} 0 & 1 & 0 & 0 \\ 0 & \frac{1}{\sqrt{1-\beta 2^2}} & 0 & -\frac{\beta 2}{\sqrt{1-\beta 2^2}} \\ 0 & 0 & 1 & 0 \\ 0 & -\frac{\beta 2}{\sqrt{1-\beta 2^2}} & 0 & \frac{1}{\sqrt{1-\beta 2^2}} \end{bmatrix}$$

$Roundtrip :=$

$$\begin{bmatrix} -\frac{\beta 2^2}{-1+\beta 2^2} & \frac{1}{\sqrt{1-\beta 2^2}} & 0 & -\frac{\beta 2}{-1+\beta 2^2} \\ \frac{\beta 2^2 (\sqrt{1-\beta 2^2}-1)}{(-1+\beta 2^2)^2} & -\frac{\beta 2^4 - 1 + \beta 2^2 \sqrt{1-\beta 2^2}}{(-1+\beta 2^2)^2} & 0 & \frac{\beta 2 (\sqrt{1-\beta 2^2}-1)}{(-1+\beta 2^2)^2} \\ 0 & 0 & 1 & 0 \\ -\frac{(\beta 2^2 \sqrt{1-\beta 2^2}-1)\beta 2}{(-1+\beta 2^2)^2} & \frac{\beta 2 (-2+2\beta 2^2+\sqrt{1-\beta 2^2})}{(-1+\beta 2^2)^2} & 0 & -\frac{\beta 2^2 \sqrt{1-\beta 2^2}-1}{(-1+\beta 2^2)^2} \end{bmatrix}$$

It is quite apparent that the round trip does not give the Identity !! Even though the in the above example, beta1=beta2 in magnitude.
Hence you do not wind up at the same position for the round trip ( a defect occurs ) (A similar result occurs for the rotation of a rigid body in 3 dimensions). The xt and yt pairs however are usually attributed to translations, not rotations, yet the compound round trip Lorentz transformations do not yield the identity. In euclidean geometry, a round trip translational displacement, first along x, then along y, then along minus x and then along minus y yields the identity. Not so in Minlowski geometry.

# Lorentz Space Space transformations

As above, there are both proper and improper space-space Lorentz transformations that preserve the Minlowski metric.

A format for the xy pair generator is given by the expression

**Lxy**

```
> Lxy := subs(Q^2=1,R^2=1,evalm(matrix([[R*cos(beta),
  Q*sin(beta),0,0], [-Q*sin(beta),R*cos(beta),0,0],[0, 0, 1, 0], [0,
  0, 0,1]])));DETLxy:=simplify(subs(Q^2=1,R^2=1,det(Lxy)));
```

$$Lxy := \begin{bmatrix} R\cos(\beta) & Q\sin(\beta) & 0 & 0 \\ -Q\sin(\beta) & R\cos(\beta) & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

$$DETLxy := 1$$

```
> `MM'`:=simplify(subs(Q^2=1,R^2=1,innerprod(transpose(Lxy),MM,Lxy))
  );
```

$$MM' := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 \end{bmatrix}$$

From experience with 3 dimensions, it is apparent that the proper space space Lorentz transformations correspond to rotations: Lxy is a rotation about the z axis. R goes to minus 1 implies for small beta that both the x and y axes get inverted. Q gors to minus Q implies the sense of rotation is changed. The Proper Lorentz space space transformations are members of the orthogonal group. transpose[Lxy] times [Lxy] = [I]

```
> `Lxysq`:=simplify(subs(Q^2=1,R^2=1,innerprod(transpose(Lxy),Lxy)))
  ;
```

$$Lxysq := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

The improper space-space Lorentz transformations are generated from

**Lxyimp**

```
> Lxyimp := subs(Q^2=1,R^2=1,evalm(matrix([[R*cos(beta),
  Q*sin(beta),0,0], [Q*sin(beta),-R*cos(beta),0,0],[0, 0, 1, 0], [0,
  0, 0,1]])));DETLxyimp:=simplify(subs(Q^2=1,R^2=1,det(Lxyimp)));
```

$$Lxyimp := \begin{bmatrix} R\cos(\beta) & Q\sin(\beta) & 0 & 0 \\ Q\sin(\beta) & -R\cos(\beta) & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

$$DETLxyimp := -1$$

To show that Lxyimp is a Lorentz transformation, compute

```
> `MM'`:=simplify(subs(Q^2=1,R^2=1,innerprod(transpose(Lxyimp),MM,Lx
  yimp)));
```

$$MM' := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 \end{bmatrix}$$

QED

Now observe that the square of an improper Lorentz space-space transformation is the identity. Hence the improper space space transformations (which reflect one or the other of the paired axes, but not both) generate involutions.

```
> `Lxyimpsq`:=simplify(subs(Q^2=1,R^2=1,innerprod(transpose(Lxyimp),
  Lxyimp)));
```

$$Lxyimpsq := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Comparing space-time and space-space Lorentz transformations it is apparent that the proper transformations can reflect both axes of the pair (or not), while improper transformations can and will always reflect only one of the elements of the pair.

To this stage, the functions beta in the above generating formulas (six different betas related to the six different paired axes)
can be functions of the independent variables, {x,y,z,t}. Hence if care is taken with differentiations of the matrix elements, it is possible to devise Lorentz transformations that can be associated with **accelerated** systems. The induced transformations on the differentials that preserves the differential line element ds^2 = <dR|[MM]|dR> may lead to a system of ordinary differential equations that are not uniquely integrable. For example, compute the vector array of differential 1-forms, sigma:

```
> sigma:=innerprod(Lxt,DR);
```

$$\sigma := \left[ \frac{R\,\mathrm{d}(x) + Q\,\beta\,\mathrm{d}(Ct)}{\sqrt{1-\beta^2}},\ \mathrm{d}(y),\ \mathrm{d}(z),\ \frac{Q\,\beta\,\mathrm{d}(x) + R\,\mathrm{d}(Ct)}{\sqrt{1-\beta^2}} \right]$$

```
>
```

The question arises: "Is each component of sigma equal to the differential of a function. In the above formula it is apparent that dY = dy and dT = dt are perfect differentials, but is sigma1 = d(X) and is sigma4 = d(T) ??? This question is subject to test. If true then the exterior differential of sigma1 and sigma4 must vansish.

```
> sigma1:=sigma[1];sigma4:=sigma[4];`d(sigma1)`:=simplify(d(sigma[1]
  ));`d(sigma4)`:=simplify(d(sigma[4]));ffx:=simplify(wcollect(sigma
  [1]&^d(sigma[1]))):fft:=simplify(wcollect(sigma[4]&^d(sigma[4]))):
```

$$\sigma 1 := \frac{R\,\mathrm{d}(x) + Q\,\beta\,\mathrm{d}(Ct)}{\sqrt{1-\beta^2}}$$

$$\sigma 4 := \frac{Q\,\beta\,\mathrm{d}(x) + R\,\mathrm{d}(Ct)}{\sqrt{1-\beta^2}}$$

$$d(sigma1) := -\frac{R\,\beta\,(\mathrm{d}(\beta)\ \&\wedge\ \mathrm{d}(x)) + Q\,(\mathrm{d}(\beta)\ \&\wedge\ \mathrm{d}(Ct))}{(-1+\beta^2)\sqrt{1-\beta^2}}$$

$$d(sigma4) := -\frac{Q\,(\mathrm{d}(\beta)\ \&\wedge\ \mathrm{d}(x)) + R\,\beta\,(\mathrm{d}(\beta)\ \&\wedge\ \mathrm{d}(Ct))}{(-1+\beta^2)\sqrt{1-\beta^2}}$$

# TO BE CONTINUED

```
> DEQ:=cos(Omega)*d(X)-d(x)-sin(Omega)*d(t);d(DEQ);FRO:=DEQ&^d(DEQ);
```

For FRO goes to zero implies X = X(x,t) which is the desired solution.

```
> GG:=evalm(simplify(inverse(FF)));
> sigma:=innerprod(FF,d(coord));
>  'd_sigma1'=d(sigma[1]);
> 'd_sigma2'=d(sigma[2]);
> 'd_sigma3'=d(sigma[3]);'d_sigma4'=d(sigma[4]);toptorsion1:=sigma[1
  ]&^d(sigma[1]);
>
```

The induced 1-forms are not exact, and not closed, but they are integrable.  If the induced 1-forms were all closed, the the fundamental group is the group of translations. (Slebodzinski p.  293)

```
> DGG:=array([[d(GG[1,1]),d(GG[1,2]),d(GG[1,3]),d(GG[1,4])],[d(GG[2,
  1]),d(GG[2,2]),d(GG[2,3]),d(GG[2,4])],[d(GG[3,1]),d(GG[3,2]),d(GG[
  3,3]),d(GG[3,4])],[d(GG[4,1]),d(GG[4,2]),d(GG[4,3]),d(GG[4,4])]]):
>
> CARTANRIGHT:=simplify(innerprod(-DGG,FF));
> CONJ:=simplify(innerprod(FF,CARTANRIGHT,-GG)):
```

The Left Cartan matrix is the Shipov Connection (IMO)

```
> CARTANLEFT:=simplify(innerprod(FF,DGG));
> simplify(evalm(CARTANLEFT-CONJ));
> beta:=V(Ct)/C;
```

**CARTANLEFT IS THE negative CONJUGATE OF CARTANRIGHT**

NOW use tensor methods

First compute the differentials of the inverse matrix

```
> for i from 1 to dim do for j from 1 to dim do for k from 1 to dim
  do d1GG[i,j,k] := (diff(GG[i,j],coord[k])) od od od:
```

Compute the elements of the matrix product of - d[G][F]

# which is the right Cartan matrix

```
> for b from 1 to dim do for a from 1 to dim do for k from 1 to dim
  do  ss:=0;for m from 1 to dim do ss := ss+(d1GG[a,m,k]*FF[m,b]);
  CC[a,b,k]:=simplify(-ss) od od od od ;
>
> for b from 1 to dim do for a from 1 to dim do for k from 1 to dim
  do if CC[a,b,k]=0 then else print(`Cabk`(a,b,k)=CC[a,b,k]) fi od
  od od ;
```

## THE non zero CARTAN CONNECTION coefficients.

## C(abk) index (1,-1,-1)

These results agree with matrix method above.

Now compute the Anti symmetric [bk] components of the Cartan connection:

```
> for j from 1 to dim do for i from 1 to dim do for k from 1 to dim
  do  ss := (CC[i,j,k]-CC[i,k,j])/2; TTCCS[i,j,k]:=ss od od od  ;
> for i from 1 to dim do for j from 1 to dim do for k from 1 to dim
  do if TTCCS[i,j,k]=0 then else
  print(`CartanaffineTorsion`(i,k,j)=TTCCS[i,k,j]) fi od od od ;
```

# If no entries appear here, there is no affine torsion

Next construct the induced metric on the initial state

Note that if W is not a function of x, y, or z, then there is no affine torsion.

**Christoffel Connection coefficients from the metric**

```
> metric:=simplify(innerprod(transpose(FF),LGUN,FF));det(metric);
> ssigma:=innerprod(metric,d(coord));
> d(ssigma[1]);d(ssigma[2]);
> d(ssigma[3]);d(ssigma[4]);toptormetric3:=wcollect(ssigma[3]&^d(ssi
  gma[3]));toptormetric4:=wcollect(ssigma[4]&^d(ssigma[4]));
>
> metricinverse:=inverse(metric):
> for i from 1 to dim do for j from 1 to dim do for k from 1 to dim
  do d1gun[i,j,k] := (diff(metric[i,j],coord[k])) od od od:
> for i from 1 to dim do for j from i to dim do for k from 1 to dim
  do C1S[i,j,k] := 0  od od od; for i from 1 to dim do for j from 1
  to dim do for k from 1 to dim do C1S[i,j,k] :=
  1/2*d1gun[i,k,j]+1/2*d1gun[j,k,i]-1/2*d1gun[i,j,k]  od od od;
>  for k from 1 to dim do for i from 1 to dim do for j from 1 to dim
  do ss := 0; for m to dim do ss := ss+metricinverse[k,m]*C1S[i,j,m]
  od; C2S[k,i,j] := simplify(factor(ss),trig) od od od;
> for i from 1 to dim do for j from 1 to dim do for k from 1 to dim
  do if C2S[i,j,k]=0 then  else print(`Gamma2`(i,j,k)=C2S[i,j,k]) fi
  od od od;
```

**The non zero Christoffel Connection coefficients 2nd kind**

## Gamma2(a,b,k) index (1,-1,-1)

```
>
```

NOTE THAT for the Jacobian basis frame, the Cartan Right matrix is EQUAL to the Christoffel CONNECTION matrix.  This conjecture works, but I have not proved it abstractly.

NOw compute the Tabk as

**Tright(abk) = Cartanright(abk) - Gamma(abk)**

```
> for i from 1 to dim do for j from 1 to dim do for k from 1 to dim
  do ss:=0; ss := (CC[i,j,k]-C2S[i,j,k]); CCTR[i,j,k]:=simplify(ss)
  od od od   ;
> for i from 1 to dim do for j from 1 to dim do for k from 1 to dim
  do if (C2S[i,j,k]=0 and CC[i,j,k]=0) then else
  print(`T`(i,j,k)=simplify(CCTR[i,j,k])) fi od od od ;
```

# The "Right Rotation Coefficients" vanish if the frame is an integrable Jacobian matrix. There is no AFFINE torsion in the sense of an anti-symmetric component of the connection.

The Anti-symmetric part of the RIGHT CARTAN MATRIX vanishes for a frame constructed from the Jacobian matrix of a map.

# Now compute the Shipov Delta connection, which is the Left Cartan matrix

```
> for a from 1 to dim do for j from 1 to dim do for k from 1 to dim
  do d1GG[a,j,k] := simplify(diff(GG[a,j],coord[k])) od od od:
```

Compute the elements of the matrixx product of [F]d[G]

```
> for i from 1 to dim do for j from 1 to dim do for k from 1 to dim
  do  ss:=0;for m to dim do ss := ss+FF[i,m]*(d1GG[m,j,k]);
  DD[i,j,k]:=simplify(ss) od od od od ;
> for i from 1 to dim do for j from 1 to dim do for k from 1 to dim
  do if DD[i,j,k]=0 then else print(`Delta`(i,j,k)=DD[i,j,k]) fi od
  od od ;
```

**NON-ZERO SHIPOV CONNECTION coefficients**

**Cartan left matrix =Delta(ijk) index (1,-1,-1)**

These values agree with the matrix methods.

The anti-symmetric part of the Shipov (Left) CARTAN Connection

```
> for j from 1 to dim do for i from 1 to dim do for k from 1 to dim
  do  ss := (DD[i,j,k]-DD[i,k,j])/2; TTS[i,j,k]:=simplify(ss) od od
  od  ;
> for i from 1 to dim do for j from 1 to dim do for k from 1 to dim
  do if TTS[i,j,k]=0 then else
  print(`ShipovTorsion`(i,k,j)=TTS[i,k,j]) fi od od od ;
```

Now compute the Tijk assuming the formula

**T(ijk) = Cartanleft(ijk) - Gamma(ijk)**

```
> for i from 1 to dim do for j from 1 to dim do for k from 1 to dim
  do ss:=0; ss := (DD[i,j,k]-C2S[i,j,k]);
  SHIPTR[i,j,k]:=simplify(ss)  od od od   ;
```

>

>

```
> for i from 1 to dim do for j from 1 to dim do for k from 1 to dim
  do if (C2S[i,j,k]=0 and DD[i,j,k]=0) then else
  print(`T`(i,j,k)=simplify(SHIPTR[i,j,k])) fi od od od ;
```

>

# The "LEFT Rotation Coefficients"

**But it does not seem to make sense (to me) to subtract the Christoffel part from Delta.**

>

>